

## Parallel Printer Routines

The INIT routine below initialises the TANEX 6522 VIA. The registers are as follows:

Data Direction Register (DDRA / DDRB)

Set all peripheral pins to output

Peripheral Control Register (PCR)

Set Control line 1 for negative active edge

Set Pulse output mode

Interrupt Enable Register (IER)

Disable all interrupts

Interrupt Flag Register (IFR)

Check Control Line 1 Flag

Table below shows the required settings depending on the VIA / Output socket used.

I/O Socket	A1	B1	C1	D1
6522	A2	A2	B2	B2
DDRA / DDRB	\$BFC3	\$BFC2	\$BFE3	\$BFE2
Initialisation value	#\$FF	#\$FF	#\$FF	#\$FF
PCR	\$BFCC	\$BFCC	\$BFEC	\$BFEC
Initialisation value	#\$0A	#\$A0	#\$0A	#\$A0
IER	\$BFCE	\$BFCE	\$BFEE	\$BFEE
Initialisation value	#\$7F	#\$7F	#\$7F	#\$7F
IFR	\$BFCD	\$BFCD	\$BFED	\$BFED
Control line 1 flag	#\$02	#\$10	#\$02	#\$10

The routines below assume the B2 VIA & D1 socket are used.

```

0001  INIT          LDA  #$FF          1A60  A9  FF          ; Initialise 6522 registers
0002                STA  $BFE2        1A62  8D  E2  BF        ; set ports to output
0003                LDA  #$A0          1A65  A9  A0
0004                STA  $BFEC        1A67  8D  EC  BF        ; set pulse output mode
0005                LDA  #$7F          1A6A  A9  7F
0006                STA  $BFEE        1A6C  8D  EE  BF        ; disable all interrupts
0007                RTS                1A67  60                ; return
    
```

The PRINT routine prints the character held in the Accumulator. It first checks to see if the character to be printed is a Carriage Return (\$13). If so, it is sent to the printer followed by a Line Feed character. Otherwise the character is sent straight to the printer and the routine exits. If the printer automatically advances the line feed on receipt of a carriage return character, then the routine should be entered at PRCHAR.

```

0008 PRINT      CMP  #$0D      1A70 C9 0D      ; check carriage return?
0009           BNE  PRCHAR    1A72 D0 08      ; no – skip to print character
0010           STA  $BFE0     1A74 8D E0 BF    ; yes – process it
0011           JSR  WAIT      1A77 20 IF 14    ; wait printer until ready
0012           LDA  #$0A      1A7A A9 0A      ; load Line Feed character
0013 PRCHAR     STA  $BFE0     1A7C 8D E0 BF    ; send character to printer
0014 WAIT       LDA  $BFED     1A7F AD ED BF    ; wait routine
0015           AND  #$10      1A82 29 10      ; check control line 1 flag
0016           BEQ  WAIT      1A84 F0 F9      ; printer not finished
0017           RTS           1A8F 60      ; return

```

The DUMP routine below dumps the Microtan screen to the printer.

```

0018 DUMP      JSR  INIT      1A87 20 60 1A    ;
0019           LDA  #$0        1A8A A9 00      ;
0020           STA  GET+1     1A8C 8D 9A 1A    ;
0021           LDA  #$2        1A8F A9 02      ;
0022           STA  $GET+2    1A91 8D 9B 1A    ;
0023 NEWLIN     JSR  CR        1A94 20 C1 1A    ; print CR
0024           LDY  #$0        1A97 A0 00      ;
0025 GET        LDA  $200,Y    1A99 B9 00 02    ;
0026           AND  #$7F      1A9C 29 7F      ;
0027           CMP  #$20      1A9E C9 20      ; alpha-numeric?
0028           BPL  SKIP      1AA0 10 02      ; yes - skip
0029           LDA  #$20      1AA2 A9 20      ; print space
0030 SKIP      JSR  PRCHAR    1AA4 20 7C 1A    ; send to printer
0031           INY           1AA7 C8          ;
0032           CPY  #$20      1AA8 C0 20      ;
0033           BNE  GET        1AAA D0 ED      ;
0034           CLC           1AAC 18          ;
0035           LDA  GET+1     1AAD AD 9A 1A    ;
0036           ADC  #$20      1AB0 69 20      ;
0037           STA  GET+1     1AB2 8D 9A 1A    ;
0038           BNE  NEWLIN    1AB5 D0 DD      ;
0039           INC  GET+2     1AB7 EE 9B 1A    ;
0040           LDA  GET+2     1ABA AD 9B 1A    ;
0041           CMP  #$4        1ABD C9 04      ; bottom of screen reached?
0042           BNE  NEWLIN    1ABF D0 D3      ; no - new line
0043 CR        LDA  #$0A      1AC1 A9 0A      ; output LF
0044           JSR  PRCHAR    1AC3 20 7C 1A    ;
0045           LDA  #$0D      1AC6 A9 0D      ; output CR
0046           JSR  PRCHAR    1AC8 20 7C 1A    ;
0047           RTS           1ACB 60      ; end

```

The FORMF routine sends a Form Feed character to the printer to eject the page.

```

0048 FORMF     LDA  #$0C      1ACC A9 0C      ; Form Feed Character
0049           BNE  PRCHAR    1ACE D0AC      ; send to printer

```